
fl_MCPTT_Info_AmbientListening_RX (new)
	  function fl_MCPTT_Info_AmbientListening_RX(charstring p_AmbientListeningType) return template (present) MCPTT_Info_Type
  {
    var template (present) MCPTT_Info_Type v_MCPTT_Info := cr_MCPTT_Info_Base;
    var charstring v_AmbientListeningTypeExt := f_MCX_McpttInfoExt_Element("ambient-listening-type", p_AmbientListeningType);
    
    v_MCPTT_Info.mcptt_Params.session_type := "ambient-listening";
    v_MCPTT_Info.mcptt_Params.anyExt.elem_list := { *, v_AmbientListeningTypeExt, * };

    return v_MCPTT_Info;
  }



fl_MCPTT_Info_AmbientListening_TX (new)
	  function fl_MCPTT_Info_AmbientListening_TX(charstring p_AmbientListeningType) return template (value) MCPTT_Info_Type
  {
    var template (value) MCPTT_Info_Type v_MCPTT_Info := cs_MCPTT_Info_Base;
    var charstring v_AmbientListeningTypeExt := f_MCX_McpttInfoExt_Element("ambient-listening-type", p_AmbientListeningType);
    
    v_MCPTT_Info.mcptt_Params.session_type := "ambient-listening";
    v_MCPTT_Info.mcptt_Params.anyExt.elem_list := { v_AmbientListeningTypeExt };

    return v_MCPTT_Info;
  }



fl_MCX_CallSetupMO_AmbientListening
	  function fl_MCX_CallSetupMO_AmbientListening(MCX_CommencementMode_Type p_CommencementMode) runs on MCX_IMS_PTC return IMS_InviteRequestWithSdp_Type
  {
    var MCX_InviteRequestWithSdpAndXml_Type v_InviteRequestWithSdpAndXml;
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var INVITE_Request v_InviteRequest;
    var MCX_MediaControlCondition_Type v_FloorControlConditionSdpOffer;
    var MCX_MediaControlConditionSdpAnswer_Type v_FloorControlConditionSdpAnswer;
    var charstring v_AmbientListeningTypeExt;
    var charstring v_AmbientListeningType;
    var template (value) IMS_RoutingInfo_Type v_RoutingInfo_DL;
    var template (value) Contact v_Contact;
    var template (value) MessageHeader v_MessageHeader200OK;
    var template (value) SDP_Message v_SDP_Answer;
    var template (value) SDP_attribute v_DirectionAttributeRX;
    var template (value) SemicolonParam_List v_FeatureParamList;
    var template (present) MCX_SdpInfoRX_Type v_SdpInfoRX;
    var template (present) MCX_MIME_XmlBodies_Type v_XmlBodies;

    select (p_CommencementMode) {
      case (AMBIENT_LISTENING_REMOTE) {
        v_AmbientListeningType := "remote-init";
        v_DirectionAttributeRX :=  cs_SDP_Attribute_recvonly;
        v_FloorControlConditionSdpOffer := floorControlWithoutImplictGrantRequest;
        v_FloorControlConditionSdpAnswer := implicitFloorRequestNoted;
      }
      case (AMBIENT_LISTENING_LOCAL) {
        v_AmbientListeningType := "local-init";
        v_DirectionAttributeRX :=  cs_SDP_Attribute_sendonly;
        v_FloorControlConditionSdpOffer := floorControlWithImplictGrantRequest;
        v_FloorControlConditionSdpAnswer := implicitFloorRequestGranted;
      }
    }
    
    // SDP info
    v_SdpInfoRX := f_MCPTT_SdpInfo_RX(PRIVATE_CALL, SDP_OFFER, v_FloorControlConditionSdpOffer, v_DirectionAttributeRX);

    // XML Bodies
    v_XmlBodies := f_MCPTT_InviteMO_XmlBodies_Init(PRIVATE_CALL);
    v_XmlBodies.McpttInfo := fl_MCPTT_Info_AmbientListening_RX(v_AmbientListeningType);
    v_AmbientListeningTypeExt := f_MCX_McpttInfoExt_Element("ambient-listening-type", v_AmbientListeningType);
    v_XmlBodies.McpttInfo.mcptt_Params.session_type := "ambient-listening";
    v_XmlBodies.McpttInfo.mcptt_Params.anyExt.elem_list := { *, v_AmbientListeningTypeExt, * };

    // Receive INVITE
    v_InviteRequestWithSdpAndXml := f_MCPTT_InviteMO_ReceiveINVITE(PRIVATE_CALL, -, p_CommencementMode, -, v_SdpInfoRX, v_XmlBodies);
    v_InviteRequestWithSdp := v_InviteRequestWithSdpAndXml.InviteRequestWithSdp;
    v_RoutingInfo_DL := f_IMS_RoutingInfo_ULtoDL(v_InviteRequestWithSdp.RoutingInfo);
    v_InviteRequest := v_InviteRequestWithSdp.Invite;

    // Send 100 Trying
    f_MCX_InviteMO_Send100Trying(v_InviteRequestWithSdp);

    // Send 200 OK
    v_Contact := f_MCX_CallSetupMO_ContactSS(MCPTT);
    v_MessageHeader200OK := f_MCX_CommonResponse_200_MessageHeaderTX(v_InviteRequest.msgHeader, v_Contact, cs_ContentTypeSDP);
    v_FeatureParamList := { cs_GenericParam("g.3gpp.mcptt.ambient-listening-call-release") };
    v_MessageHeader200OK.featureCaps := cs_FeatureCaps(v_FeatureParamList);
    v_SDP_Answer := f_MCPTT_SdpAnswer_TX(v_InviteRequestWithSdp.SdpOffer, v_FloorControlConditionSdpAnswer);
    IMS_Server.send(cas_IMS_DATA_RSP(v_RoutingInfo_DL, cs_Response(c_statusLine200, v_MessageHeader200OK, cs_MessageBody_SDP(v_SDP_Answer))));

    // Receive ACK
    f_MCX_InviteMO_ReceiveACK(-, v_InviteRequest);

    return v_InviteRequestWithSdp;
  }



fl_MCX_CallSetupMT_AmbientListening
	  function fl_MCX_CallSetupMT_AmbientListening(MCX_CommencementMode_Type p_CommencementMode) runs on MCX_IMS_PTC
  {
    var IMS_InviteRequestWithSdp_Type v_InviteRequestWithSdp;
    var charstring v_AmbientListeningTypeExt;
    var charstring v_AmbientListeningType;
    var template (value) MCX_MIME_XmlBodies_Type v_XmlBodiesTX;
    var template (value) SDP_Message v_SdpOfferTX;
    var template (value) SDP_attribute v_DirectionAttributeTX;
    var template (value) SDP_attribute v_DirectionAttributeRX;
    var template (present) MCX_SdpInfoRX_Type v_SdpInfoRX;
    var template (present) MCPTT_Info_Type v_McpttInfo;
    var boolean v_AllowTrying := true;

    select (p_CommencementMode) {
      case (AMBIENT_LISTENING_REMOTE) {
        v_AmbientListeningType := "remote-init";
        v_DirectionAttributeTX :=  cs_SDP_Attribute_recvonly;   /* @sic R5s211377 - additional change: exchange of RX/TX sic@ */
        v_DirectionAttributeRX :=  cs_SDP_Attribute_sendonly;   /* @sic R5s211377 - additional change: exchange of RX/TX sic@ */
      }
      case (AMBIENT_LISTENING_LOCAL) {
        v_AmbientListeningType := "local-init";
        v_DirectionAttributeTX :=  cs_SDP_Attribute_sendonly;   /* @sic R5s211377 - additional change: exchange of RX/TX sic@ */
        v_DirectionAttributeRX :=  cs_SDP_Attribute_recvonly;   /* @sic R5s211377 - additional change: exchange of RX/TX sic@ */
      }
    }
    
    // SDP info
    v_SdpOfferTX := f_MCPTT_SdpOffer_TX(PRIVATE_CALL);
    v_SdpOfferTX.media_list[0].attributes := f_SDP_Attributes_Add_TX(v_SdpOfferTX.media_list[0].attributes, v_DirectionAttributeTX);

    // XML Bodies
    v_XmlBodiesTX := f_MCPTT_InviteMT_XmlBodies_Init(PRIVATE_CALL);
    v_XmlBodiesTX.McpttInfo := fl_MCPTT_Info_AmbientListening_TX(v_AmbientListeningType);
    v_AmbientListeningTypeExt := f_MCX_McpttInfoExt_Element("ambient-listening-type", v_AmbientListeningType);
    v_XmlBodiesTX.McpttInfo.mcptt_Params.session_type := "ambient-listening";
    v_XmlBodiesTX.McpttInfo.mcptt_Params.anyExt.elem_list := { v_AmbientListeningTypeExt };

    // Send INVITE
    v_InviteRequestWithSdp := f_MCX_InviteMT_SendINVITE(MCPTT, PRIVATE_CALL, -, p_CommencementMode, -, -, v_SdpOfferTX, v_XmlBodiesTX);

    // receive 100 Trying, 200 OK, send ACK
    v_SdpInfoRX := f_MCPTT_SdpInfo_RX(PRIVATE_CALL, SDP_ANSWER, -, v_DirectionAttributeRX);
    v_McpttInfo := f_MCX_CallControl_MCPTTInfo_RX(PRIVATE_CALL, INVITE_RSP);
    f_MCPTT_InviteMT_Receive200OK_SendACK(PRIVATE_CALL, v_InviteRequestWithSdp, v_AllowTrying, -, v_SdpInfoRX, v_McpttInfo);
  }




